<commit_before># CMake version > 3.8.2
# This requirement is for the "cxx_std_11" variable in target_compile_features().
cmake_minimum_required(VERSION 3.8.2)
project(libmorton CXX)

option(LIBMORTON_BUILD_TESTS "Build unit tests for libmorton" ON)

add_library(libmorton INTERFACE)
target_compile_features(libmorton
  INTERFACE
    cxx_std_11
  )
target_compile_options(libmorton
  INTERFACE
    $<$<NOT:$<CXX_COMPILER_ID:MSVC>>:-Wall -Wextra -Wpedantic>
    $<$<CXX_COMPILER_ID:MSVC>:/W3>
  )
target_compile_definitions(libmorton
  INTERFACE
    # To avoid errors related to std::min/std::max
    $<$<CXX_COMPILER_ID:MSVC>:NOMINMAX>
  )
target_include_directories(libmorton
  INTERFACE
    libmorton/include
  )
add_subdirectory(libmorton)
add_library(libmorton::libmorton ALIAS libmorton)

if (LIBMORTON_BUILD_TESTS)
    add_subdirectory(test)
endif()<commit_msg>Add compile options for Intel C++ compiler<commit_after># CMake version >= 3.8.2
# This requirement is for the "cxx_std_11" variable in target_compile_features().
cmake_minimum_required(VERSION 3.8.2)
project(libmorton CXX)

option(LIBMORTON_BUILD_TESTS "Build unit tests for libmorton" ON)

add_library(libmorton INTERFACE)
target_compile_features(libmorton
  INTERFACE
    cxx_std_11
  )
target_compile_options(libmorton
  INTERFACE
    # gcc and clang 
    $<$<OR:$<CXX_COMPILER_ID:GNU>,$<CXX_COMPILER_ID:Clang>>:-Wall -Wextra -Wpedantic>
    # Intel compiler
    $<$<CXX_COMPILER_ID:Intel>:$<IF:$<PLATFORM_ID:Windows>,/W3,-w3>>
    # MSVC
    $<$<CXX_COMPILER_ID:MSVC>:/W3>
  )
target_compile_definitions(libmorton
  INTERFACE
    # MSVC
    # To avoid errors related to std::min/std::max
    $<$<CXX_COMPILER_ID:MSVC>:NOMINMAX>
  )
target_include_directories(libmorton
  INTERFACE
    libmorton/include
  )
add_subdirectory(libmorton)
add_library(libmorton::libmorton ALIAS libmorton)

if (LIBMORTON_BUILD_TESTS)
    add_subdirectory(test)
endif()
{'fraction_non_alphanumeric': 0.10427528675703858, 'fraction_numerical': 0.01251303441084463, 'mean_word_length': 5.853571428571429, 'pattern_counts': {'":': 0, '<': 22, '<?xml version=': 0, '>': 24, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2302778', 'n_tokens_mistral': 793, 'n_tokens_neox': 738, 'n_words': 123}